

Tip Controls

Objekat Controls vam omogućava da dodate kontrole u Grafički prozor, kao i da ih pomerate i interagujete sa njima.

U normalnim slučajevima, trebalo bi da izbegavate korišćenje ovog tipa, jer je to Small Basic nasleđeni tip, i da umesto toga koristite sVB dizajner obrazaca za dodavanje kontrola u vaše obrasce.

Tip Controls ima sledeće članove:

Controls.AddButton(caption, left, top) As Button

Dodaje novu kontrolu Button u grafički prozor tokom izvršavanja.


Parametri:

- *caption*: Tekst koji će se prikazati na dugmetu.
- *left*: X-pozicija kontrole.
- *top*: Y-pozicija kontrole.

Vraća: Taster dugmeta. sVB može da obrađuje ovaj taster kao objekat tipa Button, tako da možete pristupiti svojstvima, metodama i događajima Button preko njega.

Primer: U sVB-u, pređite na karticu koda i otvorite novu Small Basic datoteku (klikom na dugme New na traci sa alatkama ili pritiskom na Ctrl+N). Dodajte ovaj kôd u novu datoteku:

```
BtnOK = Controls.AddButton("OK", 10, 20)
BtnOK.Width = 100
BtnOK.Height = 50
BtnOK.FontSize = 14
BtnOK.OnClick = BtnOK_OnClick
Sub BtnOK_OnClick()
Forms.ShowMessage("Hello", "BtnOK")
EndSub
```



Pritisnite F5 da biste pokrenuli program. Dugme OK će biti prikazano u grafičkom prozoru, a kada kliknete na njega, prikazaće se prozor sa porukom „Hello“:

Controls.AddCheckBox(caption, left, top, checked) As CheckBox

Dodaje novu kontrolu CheckBox u grafički prozor.


Parametri:

- *caption*: Tekst koji će se prikazati na CheckBox-u.
- *left*: X-pozicija kontrole.
- *top*: Y-pozicija kontrole.
- *checked*: Vrednost koju treba podesiti za svojstvo CheckBox.Checked.

Vraća: Ključ CheckBox-a. sVB može da obrađuje ovaj ključ kao objekat tipa CheckBox, tako da preko njega možete pristupiti svojstvima, metodama i događajima CheckBox-a.

Primer: U sVB-u, pređite na karticu koda i otvorite novu malu osnovnu datoteku (klikom na dugme Novo na traci sa alatkama ili pritiskom na Ctrl+N). Dodajte ovaj kôd u novu datoteku:

```
' Napravi polje za potvrdu i obradite njegov događaj OnCheck:
CheckBox1 = Controls.AddCheckBox(
"Show Label", 10, 50, True)
CheckBox1.OnCheck = CheckBox1_OnCheck
' Napravi labelu levo od polja za potvrdu:
Label1 = Controls.AddLabel(
"Zdravo svete!",
```



```

CheckBox1.Left + CheckBox1.Width + 30,
CheckBox1.Top)
' Podesi svojstva Labela:
Label1.BackColor = Colors.AliceBlue
' Poravnaj sredinu labela i polja za potvrdu horizontalno:
Label1.Top = Label1.Top -
(Label1.Height - CheckBox1.Height) / 2
' Ručovalac OnCheck sakriva ili prikazuje Labelu:
Sub CheckBox1_OnCheck()
Label1.Visible = CheckBox1.Checked
EndSub

```

Pritisnite F5 da biste pokrenuli program. Grafički prozor će prikazati polje za potvrdu i labelu. Označite i uklonite oznaku iz polja za potvrdu da biste prikazali ili sakrili labelu.

Controls.AddComboBox(left, top, width, height) As ListBox

Dodaje novu kontrolu ComboBox u grafički prozor.

Parametri:

- *left*: X-pozicija kontrole.
- *top*: Y-pozicija kontrole.
- *width*: Širina kontrole. Koristite -1 za automatsku širinu.
- *height*: Visina kontrole. Koristite -1 za automatsku visinu.


Vraća: Ključ kombinovanog okvira. sVB može da obrađuje ovaj ključ kao objekat tipa ComboBox, tako da možete koristiti svojstva, metode i događaje ComboBox-a preko njega.

Primer: U sVB-u, pređite na karticu koda i otvorite novu malu osnovnu datoteku (klikom na dugme Novo na traci sa alatima ili pritiskom na Ctrl+N). Dodajte ovaj kôd u novu datoteku:

```

ComboBox1 = Controls.AddComboBox(100, 20, 150, -1)
ComboBox1.AddItem({"a", "b", "c"})
ComboBox1.OnSelection = ComboBox1_Select
ComboBox1.SelectedIndex = 3
Sub ComboBox1_Select()
Print(ComboBox1.SelectedItem)
EndSub
Sub Print(item)
' Obriši bilo koji prethodno napisani tekst
c = GraphicsWindow.BackgroundColor
GraphicsWindow.BrushColor = c
GraphicsWindow.FillRectangle(
100, 45,
GraphicsWindow.Width - 105,
GraphicsWindow.Height - 40
)
' Upiši izabranu stavku:
GraphicsWindow.BrushColor = Colors.Black
GraphicsWindow.DrawText(
105, 50,
Text.Format(
"Izabrana stavka je [1][2][1]",
{Chars.Quote, item}
)
)
EndSub

```



Pritisnite F5 da biste pokrenuli program. Grafički prozor će prikazati kombinovani okvir sa poslednjom izabranom stavkom, a izabrana stavka će biti zapisana ispod kombinovanog okvira i biće ažurirana svaki put kada promenite izabranu stavku.

Controls.AddDatePicker(left, top, width, selectedDate) As DatePicker

Dodaje novu kontrolu DatePicker u grafički prozor.


Parametri:

- *left*: X-pozicija kontrole.
- *top*: Y-pozicija kontrole.
- *width*: Širina kontrole. Koristite -1 za automatsku širinu.
- *selectedDate*: Datum koji će biti izabran u kontroli kada se prikaže.

Vraća: Ključ kontrole DatePicker. sVB može da obrađuje ovaj ključ kao objekat tipa DatePicker, tako da možete pristupiti svojstvima, metodama i događajima DatePicker-a preko njega.

Primer: U sVB-u, pređite na karticu koda i otvorite novu malu osnovnu datoteku (klikom na dugme Novo na traci sa alatkama ili pritiskom na Ctrl+N). Dodajte ovaj kôd u novu datoteku:

```
DatePicker1 = Controls.AddDatePicker(  
75, 10, 200, #12/18/2022#)  
DatePicker1.OnSelection = DatePicker1_OnSelection  
Sub DatePicker1_OnSelection()  
Print(DatePicker1.SelectedDate)  
EndSub
```



Takođe dodajte potprogram Print koji je korišćen u prethodnom primeru.

Pritisnite F5 da biste pokrenuli program. Svaki put kada promenite izabrani datum, on će biti prikazan ispod birača datuma.

Controls.AddLabel(caption, left, top) As Label

Dodaje novu kontrolu Label u grafički prozor.

Parametri:

- *caption*: tekst koji će se prikazati na etiketi.
- *left*: X-pozicija kontrole.
- *top*: Y-pozicija kontrole.

Vraća: Ključ etikete. sVB može da obrađuje ovaj taster kao objekat tipa Label, tako da preko njega možete pristupiti svojstvima, metodama i događajima Label.

Primer: U sVB-u, pređite na karticu koda i otvorite novu malu osnovnu datoteku (klikom na dugme Novo na traci sa alatkama ili pritiskom na Ctrl+N). Dodajte ovaj kôd u novu datoteku:

```
Label1 = Controls.AddLabel("Zdravo!", 10, 20)  
Label1.OnMouseEnter = Label1_OnMouseEnter  
Label1.OnMouseLeave = Label1_OnMouseleave  
Sub Label1_OnMouseEnter()  
Label1.BackColor = Colors.AliceBlue  
EndSub  
Sub Label1_OnMouseleave()  
Label1.BackColor = Colors.Transparent  
EndSub
```

Pritisnite F5 da biste pokrenuli program. Grafički prozor će prikazati oznaku „Zdravo“, i kada pokazivač miša uđe u njegovu oblast, boja njegove pozadine će se promeniti, gde možete potvrditi da se njegova širina i visina uklapaju u sadržaj (jer smo koristili -1 za oba). Kada pokazivač miša napusti oblast labele, boja njegove pozadine će ponovo biti providna. To možete ponoviti onoliko puta koliko želite!

Controls.AddListBox(left, top, width, height) As ListBox

Dodaje novu kontrolu ListBox u grafički prozor.


Parametri:

- *left*: X-pozicija kontrole.
- *top*: Y-pozicija kontrole. - 371 -
- *width*: Širina kontrole. Koristite -1 za automatsku širinu.
- *height*: Visina kontrole. Koristite -1 za automatsku visinu.

Vraća: Ključ liste. sVB može da obrađuje ovaj ključ kao objekat tipa ListBox, tako da možete pristupiti svojstvima, metodama i događajima ListBox-a preko njega.

Primer: U sVB-u, pređite na karticu koda i otvorite novu malu osnovnu datoteku (klikom na dugme Novo na traci sa alatkama ili pritiskom na Ctrl+N). Dodajte ovaj kôd u novu datoteku:

```
ListBox1 = Controls.AddListBox(100, 20, 150, -1)
ListBox1.AddItem({"a", "b", "c"})
ListBox1.SelectedIndex = 3
```



Pritisnite F5 da biste pokrenuli program. Grafički prozor će prikazati listu sa poslednjom izabranom stavkom.

Controls.AddProgressBar(left, top, width, height, minimum, maximum) As ProgressBar

Dodaje novu kontrolu ProgressBar u grafički prozor.


Parametri:

- *left*: X-pozicija kontrole.
- *top*: Y-pozicija kontrole.
- *width*: Širina kontrole. Koristite -1 za automatsku širinu.
- *height*: Visina kontrole. Koristite -1 za automatsku visinu.
- *minimum*: Minimalna vrednost napretka
- *maximum*: Maksimalna vrednost napretka. Koristite 0 ako je maksimalna vrednost neodređena. Ovo će prikazati beskonačno pokretni okvir.

Vraća: Ključ ProgressBar-a. sVB može da obrađuje ovaj ključ kao objekat tipa ProgressBar, tako da možete pristupiti svojstvima, metodama i događajima ProgressBar-a preko njega.

Primer: U sVB-u, pređite na karticu koda i otvorite novu malu osnovnu datoteku (klikom na dugme Novo na traci sa alatkama ili pritiskom na Ctrl+N). Dodajte ovaj kôd u novu datoteku:

```
ProgressBar1 = Controls.AddProgressBar(
35, 55, 250, 30,
0, 0)
```



Pritisnite F5 da biste pokrenuli program. Grafički prozor će prikazati traku napretka sa beskrajno pokretnim okvirom.

Controls.AddRadioButton(caption, left, top, groupName, checked) As RadioButton

Dodaje novu kontrolu RadioButton u grafički prozor.

Parametri:


- *caption*: Tekst koji će se prikazati na RadioButton-u
- *left*: X-pozicija kontrole.
- *top*: Y-pozicija kontrole.
- *groupName*: Ime grupe kojoj treba dodati dugme.

- *checked*: Vrednost koju treba postaviti na svojstvo Checked

Vraća: Ključ RadioButton-a. sVB može da obrađuje ovaj ključ kao objekat tipa RadioButton, tako da možete pristupiti svojstvima, metodama i događajima RadioButton-a preko njega.

Primer: U sVB-u, pređite na karticu koda i otvorite novu malu osnovnu datoteku (klikom na dugme Novo na traci sa alatkama ili pritiskom na Ctrl+N). Dodajte ovaj kôd u novu datoteku: Kreira 2 grupe radio dugmadi, svaka sadrži dva dugmeta. Takođe dodaje oznaku za prikaz izabrane pozicije:

```
' Radio dugmad grupa1
RdoTop = Controls.AddRadioButton(
"Vrh", 35, 30, "grupa1", False)
RdoBottom = Controls.AddRadioButton(
"Dno", 235, 30, "grupa1", True)
' Radio dugmad grupa2
RdoLeft = Controls.AddRadioButton(
"Levo", 35, 80, "grupa2", True)
RdoRight = Controls.AddRadioButton(
"Desno", 235, 80, "grupa2", False)
LblPos = Controls.AddLabel("", 35, 120)
' Samo jedan obrađivač događaja za svu radio dugmad:
RdoTop.OnCheck = OnCheck
RdoBottom.OnCheck = OnCheck
RdoLeft.OnCheck = OnCheck
RdoRight.OnCheck = OnCheck
' Pozovi funkciju jednom da biste obrađivali početna radio dugmad
' stanja koja su podešena metodom AddRadioButton
OnCheck()
Sub OnCheck()
pos = ""
If RdoTop.Checked Then
pos = "Vrh"
Else
pos = "Dno"
EndIf
If RdoLeft.Checked Then
pos = pos + " Levo"
Else
pos = pos + " Desno"
EndIf
LblPos.Text = pos
EndSub
```



Pritisnite F5 da biste pokrenuli program. Grafički prozor će prikazati 4 radio dugmad sa izabranim radio dugmadima „Dno“ i „Levo“. Labela će prikazivati izabranu poziciju i ažuriraće se svaki put kada promenite izabrana radio dugmad.

Controls.AddScrollBar(left, top, width, height, minimum, maximum, value) As ScrollBar

Dodaje novu kontrolu ScrollBar u grafički prozor.

Parametri:


- *left*: X-pozicija kontrole.
- *top*: Y-pozicija kontrole. - 376 -
- *width*: Širina kontrole. Koristite -1 za automatsku širinu.
- *height*: Visina kontrole. Koristite -1 za automatsku visinu.
- *minimum*: Minimalna vrednost trake za skrolovanje.
- *maksimum*: Maksimalna vrednost trake za skrolovanje.
- *value*: Trenutna vrednost trake za skrolovanje.

Vraća: Ključ trake za skrolovanje. sVB može da obrađuje ovaj ključ kao objekat tipa

ScrollBar, tako da možete pristupiti svojstvima, metodama i događajima trake za skrolovanje preko njega.

Primer: U sVB-u, pređite na karticu koda i otvorite novu malu osnovnu datoteku (klikom na dugme Novo na traci sa alatkama ili pritiskom na Ctrl+N). Dodajte ovaj kôd u novu datoteku:

```
ScrollBar1 = Controls.AddScrollBar(  
260, -175, -1, 500,  
0, 200, 50  
)  
' Rotiraj traku za skrolovanje da bude horizontalna  
ScrollBar1.Angle = -90  
ScrollBar1.OnScroll = ScrollBar1_OnScroll  
LblValue = Controls.AddLabel("50", 520, 65)  
Sub ScrollBar1_OnScroll()  
LblValue.Text = Math.Round(ScrollBar1.Value)  
EndSub
```



Pritisnite F5 da biste pokrenuli program. Grafički prozor će prikazati traku za skrolovanje, a oznaka će pokazati njenu vrednost (što je naznačeno položajem palca). Oznaka će se ažurirati svaki put kada promenite položaj palca na traci za skrolovanje.

Controls.AddSlider(left, top, width, height, minimum, maximum, value, tickFrequency) As Slider

Dodaje novu kontrolu klizača u grafički prozor.


Parametri:

- *left*: X-pozicija kontrole.
- *top*: Y-pozicija kontrole.
- *width*: Širina kontrole. Koristite -1 za automatsku širinu.
- *height*: Visina kontrole. Koristite -1 za automatsku visinu.
- *minimum*: Minimalna vrednost klizača.
- *maximum*: Maksimalna vrednost klizača.
- *value*: Trenutna vrednost klizača.
- *tickFrequency*: Rastojanje između oznaka slajda.

Vraća: Ključ klizača. sVB može da obrađuje ovaj ključ kao objekat tipa Klizač, tako da možete pristupiti svojstvima, metodama i događajima klizača preko njega.

Primer: U sVB-u, pređite na karticu koda i otvorite novu malu osnovnu datoteku (klikom na dugme Novo na traci sa alatkama ili pritiskom na Ctrl+N). Dodajte ovaj kôd u novu datoteku:

```
Slider1 = Controls.AddSlider(  
20, 50, 500, 30,  
0, 100, 50, 10  
)  
Slider1.ForeColor = Colors.Red  
Slider1.SnapToTick = True  
Slider1.OnSlide = Slider1_OnSlide  
LblValue = Controls.AddLabel("50", 520, 65)  
Sub Slider1_OnSlide()  
LblValue.Text = Math.Round(Slider1.Value)  
EndSub
```



Pritisnite F5 da biste pokrenuli program. Grafički prozor će prikazati klizač, a labela će pokazati njegovu vrednost (što je naznačeno položajem palca). Oznaka će se ažurirati svaki put kada pomerite palac na klizaču.

Controls.AddMultiLineTextBox(left, top)

Dodaje višeredni tekstualni okvir u grafički prozor na navedenoj poziciji, što omogućava korisniku da unosi tekst u nove redove pritiskom na Enter sa tastature.

Imajte na umu da je sVB zadržao ovu metodu zbog kompatibilnosti sa Small Basic-om, ali njen posao može da obavi metoda Controls.AddTextBox, gde možete da podesite svojstvo Multi-Line tekstualnog okvira na True, kao što ćemo pokazati u sledećem primeru.

Parametri:

- *left*: X koordinata tekstualnog okvira.
- *top*: Y koordinata tekstualnog okvira.

Vraća: Ključ TextBox-a. sVB može da obrađuje ovaj ključ kao objekat tipa TextBox, tako da možete pristupiti svojstvima, metodama i događajima TextBox-a preko njega.

Controls.AddTextBox(left, top) As TextBox

Dodaje novu kontrolu TextBox u grafički prozor.


Parametri:

- *left*: X-pozicija kontrole.
- *top*: Y-pozicija kontrole.

Vraća: Ključ TextBox-a. sVB može da obrađuje ovaj ključ kao objekat tipa TextBox, tako da možete pristupiti svojstvima, metodama i događajima TextBox-a preko njega.

Primer: U sVB-u, pređite na karticu koda i otvorite novu malu osnovnu datoteku (klikom na dugme Novo na traci sa alatkama ili pritiskom na Ctrl+N). Dodajte ovaj kôd u novu datoteku:

```
TextBox1 = Controls.AddTextBox(30, 10)
TextBox1.Text = "Zdravo"
' Napravi tekst višerednim
TextBox1.MultiLine = True
TextBox1.Height = 100
```



Pritisnite F5 da biste pokrenuli program. Grafički prozor će prikazati textbox kao što je prikazano na slici:

Controls.AddTimer(interval) As WinTimer

Dodaje novu WinTimer kontrolu grafičkom prozoru tokom izvršavanja programa. Ovo vam omogućava da kreirate onoliko tajmera koliko vam je potrebno, što nije moguće kada koristite Small Basic Timer.


Parametri:

- *interval*: Vreme kašnjenja u milisekundama između otkucaja.

Vraća: Ključ tajmera. sVB može da obrađuje ovaj ključ kao objekat tipa WinTimer, tako da preko njega možete pristupiti svojstvima, metodama i događajima WinTimer-a.

Primer: U sVB-u, pređite na karticu koda i otvorite novu small basic datoteku (klikom na dugme Novo na traci sa alatkama ili pritiskom na Ctrl+N). Dodajte ovaj kôd u novu datoteku:

```
T1 = Controls.AddTimer(1000)
T1.OnTick = T1_OnTick
T2 = Controls.AddTimer(100)
T2.OnTick = T2_OnTick
Lb11 = Controls.AddLabel("", 10, 10)
Lb12 = Controls.AddLabel("", 230, 10)
T1_OnTick()
Sub T1_OnTick()
```



```
Lb11.Text = Date.Now
EndSub
Sub T2_OnTick()
Lb12.Text = Date.GetMillisecond(Date.Now)
EndSub
```

Pritisnite F5 da biste pokrenuli program. Grafički prozor će prikazati dve labelle, prva prikazuje trenutni datum i vreme i ažurira je prvi tajmer svake sekunde, dok druga prikazuje trenutne milisekunde i ažurira se svake 0,1 sekunde:

Controls.AddToggleButton(caption, left, top, width, height) As ToggleButton

Dodaje novu kontrolu ToggleButton u grafički prozor.


Parametri:

- *caption*: Tekst koji će se prikazati na dugmetu za prebacivanje.
- *left*: X-pozicija kontrole.
- *top*: Y-pozicija kontrole.
- *width*: Širina kontrole. Koristite -1 za automatsku širinu.
- *height*: Visina kontrole. Koristite -1 za automatsku visinu.

Vraća: Ključ ToggleButton-a. sVB može da obrađuje ovaj taster kao objekat tipa ToggleButton, tako da možete pristupiti svojstvima, metodama i događajima ToggleButton-a preko njega.

Primer: U sVB-u, pređite na karticu koda i otvorite novu malu osnovnu datoteku (klikom na dugme Novo na traci sa alatka ili pritiskom na Ctrl+N). Dodajte ovaj kôd u novu datoteku:

```
ToggleButton1 = Controls.AddToggleButton(
"OK", 60, 50, 150, 50)
ToggleButton1.ForeColor = Colors.Blue
ToggleButton1.OnCheck = ToggleButton1_OnCheck
ToggleButton1.Checked = True
Sub ToggleButton1_OnCheck()
ToggleButton1.FontBold = ToggleButton1.Checked
EndSub
```



Pritisnite F5 da biste pokrenuli program. Grafički prozor će prikazati dugme za prebacivanje i biće označeno. Kliknite na njega da biste promenili njegovo stanje označenog i primetite da se njegova debljina fonta menja iz podebljanog (bold) u normalnu i obrnuto kako se menja njegovo stanje označenog.

Controls.ButtonClicked

Pokreće događaj kada se klikne na bilo koju kontrolu dugmeta. Ovo je Small Basic-ov zastareli događaj i trebalo bi da obrađujete događaj Button.OnClick za svako dugme pojedinačno.

Controls.GetButtonCaption(buttonName)

Dobija trenutni natpis navedenog dugmeta. Ovo je Small Basic-ov zastareli metod i trebalo bi da koristite svojstvo Button.Text umesto njega.

Parametar:

- *buttonName*: Dugme čiji se natpis zahteva.

Vraća: Trenutni natpis dugmeta.

Controls.GetTextBoxText(textBoxName)

Dobija trenutni tekst navedenog TextBox-a. Ovo je Small Basic metoda i umesto toga treba da koristite svojstvo TextBox.Text.

Parametar:

- *textBoxName*: TextBox čiji se tekst traži.

Vraća: Tekst u TextBox-u.

Controls.HideControl(controlName)

Skriva već dodatu kontrolu. Ovo je Small Basic metoda i umesto toga treba da podesite svojstvo Visible kontrole na False.

Parametar:

- *controlName*: Ime kontrole.

Controls.LastClickedButton

Dobija poslednje dugme na koje je kliknuto u Grafičkom prozoru.

Controls.LastTypedTextBox

Dobija poslednje TextBox u koje je tekst unet.

Controls.Move(control, x, y)

Pomera kontrolu sa navedenim imenom na novu poziciju.

Parametri:

- *control*: Ime kontrole koju treba pomeriti.

- *x*: X koordinata nove pozicije.

- *y*: Y koordinata nove pozicije.

Controls.Remove(controlName)

Uklanja datu kontrolu iz grafičkog prozora.

Parametar:

- *controlName*: Ime kontrole koju treba ukloniti.

Primer: U sVB-u, pređite na karticu koda i otvorite novu datoteku Small Basic (klikom na dugme Novo na traci sa alatkama ili pritiskom na Ctrl+N). Dodajte ovaj kôd u novu datoteku:

```
Btn = Controls.AddButton("test", 10, 10)
Btn.OnClick = Click
Sub Click()
Controls.Remove(Btn)
EndSub
```

Pritisnite F5 da biste pokrenuli program. Grafički prozor će prikazati dugme, a kada kliknete, ono će biti uklonjeno!

Controls.SetButtonCaption(buttonName, caption)

Postavlja natpis navedenog dugmeta. Ovo je Small Basic metoda i umesto toga bi trebalo da koristite svojstvo Button.Text.

Parametri:

- *buttonName*: Dugme čiji natpis treba podesiti.
- *caption*: Novi natpis za dugme.

Controls.SetSize(control, width, height)

Postavlja veličinu kontrole.

Parametri:

- *control*: Naziv kontrole kojoj treba promeniti veličinu.
- *width*: Širina kontrole.
- *height*: Visina kontrole.

Controls.SetTextBoxText(textBoxName, text)

Postavlja tekst navedenog TextBox-a. Ovo je Small Basic metoda zastarelog tipa i umesto nje treba koristiti svojstvo TextBox.Text.

Parametri:

- *textBoxName*: TextBox čiji tekst treba podesiti.
- *text*: Novi tekst za TextBox.

Controls.ShowControl(controlName)

Prikazuje prethodno skrivenu kontrolu. Ovo je Small Basic metoda zastarelog tipa i umesto nje treba podesiti svojstvo Visible kontrole na True.

Parametar:

- *controlName*: Naziv kontrole.

Controls.TextTyped

Pokreće događaj kada se tekst unese u bilo koju kontrolu TextBox. Ovo je Small Basic nasleđeni događaj i trebalo bi da obradite događaj TextBox.OnTextChanged za svako tekstualno polje umesto toga.